# $Id: PKGBUILD 177597 2013-02-07 20:44:27Z stephane $
# Maintainer: Tobias Powalowski <tpowa@archlinux.org>
pkgname=seabios
pkgver=1.7.2
pkgrel=1
pkgdesc="A 16-bit x86 bios"
arch=(any)
license=('GPL3' 'LGPL3')
url="http://www.coreboot.org/SeaBIOS"
makedepends=('iasl' 'python2')
source=(http://code.coreboot.org/p/seabios/downloads/get/${pkgname}-${pkgver}.tar.gz)
options=(!strip)
md5sums=('87a1b217a568d77904accb0a43535d6b')

build()
{
    cd "${srcdir}/${pkgname}-${pkgver}"
    sed -i -e 's/python/python2/g' Makefile
    make
}

package()
{
    install -D -m644  "${srcdir}/${pkgname}-${pkgver}/out/bios.bin" "${pkgdir}/usr/share/qemu/bios.bin"
}

